// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC1155/IERC1155.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC1155/IERC1155.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC1155 compliant contract, as defined in the
 * https://eips.ethereum.org/EIPS/eip-1155[EIP].
 *
 * _Available since v3.1._
 */
interface IERC1155 is IERC165 {
    /**
     * @dev Emitted when `value` tokens of token type `id` are transferred from `from` to `to` by `operator`.
     */
    event TransferSingle(address indexed operator, address indexed from, address indexed to, uint256 id, uint256 value);

    /**
     * @dev Equivalent to multiple {TransferSingle} events, where `operator`, `from` and `to` are the same for all
     * transfers.
     */
    event TransferBatch(
        address indexed operator,
        address indexed from,
        address indexed to,
        uint256[] ids,
        uint256[] values
    );

    /**
     * @dev Emitted when `account` grants or revokes permission to `operator` to transfer their tokens, according to
     * `approved`.
     */
    event ApprovalForAll(address indexed account, address indexed operator, bool approved);

    /**
     * @dev Emitted when the URI for token type `id` changes to `value`, if it is a non-programmatic URI.
     *
     * If an {URI} event was emitted for `id`, the standard
     * https://eips.ethereum.org/EIPS/eip-1155#metadata-extensions[guarantees] that `value` will equal the value
     * returned by {IERC1155MetadataURI-uri}.
     */
    event URI(string value, uint256 indexed id);

    /**
     * @dev Returns the amount of tokens of token type `id` owned by `account`.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function balanceOf(address account, uint256 id) external view returns (uint256);

    /**
     * @dev xref:ROOT:erc1155.adoc#batch-operations[Batched] version of {balanceOf}.
     *
     * Requirements:
     *
     * - `accounts` and `ids` must have the same length.
     */
    function balanceOfBatch(address[] calldata accounts, uint256[] calldata ids)
        external
        view
        returns (uint256[] memory);

    /**
     * @dev Grants or revokes permission to `operator` to transfer the caller's tokens, according to `approved`,
     *
     * Emits an {ApprovalForAll} event.
     *
     * Requirements:
     *
     * - `operator` cannot be the caller.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns true if `operator` is approved to transfer ``account``'s tokens.
     *
     * See {setApprovalForAll}.
     */
    function isApprovedForAll(address account, address operator) external view returns (bool);

    /**
     * @dev Transfers `amount` tokens of token type `id` from `from` to `to`.
     *
     * Emits a {TransferSingle} event.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - If the caller is not `from`, it must have been approved to spend ``from``'s tokens via {setApprovalForAll}.
     * - `from` must have a balance of tokens of type `id` of at least `amount`.
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155Received} and return the
     * acceptance magic value.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 id,
        uint256 amount,
        bytes calldata data
    ) external;

    /**
     * @dev xref:ROOT:erc1155.adoc#batch-operations[Batched] version of {safeTransferFrom}.
     *
     * Emits a {TransferBatch} event.
     *
     * Requirements:
     *
     * - `ids` and `amounts` must have the same length.
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155BatchReceived} and return the
     * acceptance magic value.
     */
    function safeBatchTransferFrom(
        address from,
        address to,
        uint256[] calldata ids,
        uint256[] calldata amounts,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/token/ERC1155/IERC1155Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC1155/IERC1155Receiver.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev _Available since v3.1._
 */
interface IERC1155Receiver is IERC165 {
    /**
     * @dev Handles the receipt of a single ERC1155 token type. This function is
     * called at the end of a `safeTransferFrom` after the balance has been updated.
     *
     * NOTE: To accept the transfer, this must return
     * `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))`
     * (i.e. 0xf23a6e61, or its own function selector).
     *
     * @param operator The address which initiated the transfer (i.e. msg.sender)
     * @param from The address which previously owned the token
     * @param id The ID of the token being transferred
     * @param value The amount of tokens being transferred
     * @param data Additional data with no specified format
     * @return `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))` if transfer is allowed
     */
    function onERC1155Received(
        address operator,
        address from,
        uint256 id,
        uint256 value,
        bytes calldata data
    ) external returns (bytes4);

    /**
     * @dev Handles the receipt of a multiple ERC1155 token types. This function
     * is called at the end of a `safeBatchTransferFrom` after the balances have
     * been updated.
     *
     * NOTE: To accept the transfer(s), this must return
     * `bytes4(keccak256("onERC1155BatchReceived(address,address,uint256[],uint256[],bytes)"))`
     * (i.e. 0xbc197c81, or its own function selector).
     *
     * @param operator The address which initiated the batch transfer (i.e. msg.sender)
     * @param from The address which previously owned the token
     * @param ids An array containing ids of each token being transferred (order and length must match values array)
     * @param values An array containing amounts of each token being transferred (order and length must match ids array)
     * @param data Additional data with no specified format
     * @return `bytes4(keccak256("onERC1155BatchReceived(address,address,uint256[],uint256[],bytes)"))` if transfer is allowed
     */
    function onERC1155BatchReceived(
        address operator,
        address from,
        uint256[] calldata ids,
        uint256[] calldata values,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC1155/utils/ERC1155Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC1155/utils/ERC1155Receiver.sol)

pragma solidity ^0.8.0;

import "../IERC1155Receiver.sol";
import "../../../utils/introspection/ERC165.sol";

/**
 * @dev _Available since v3.1._
 */
abstract contract ERC1155Receiver is ERC165, IERC1155Receiver {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return interfaceId == type(IERC1155Receiver).interfaceId || super.supportsInterface(interfaceId);
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Counters.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Counters.sol)

pragma solidity ^0.8.0;

/**
 * @title Counters
 * @author Matt Condon (@shrugs)
 * @dev Provides counters that can only be incremented, decremented or reset. This can be used e.g. to track the number
 * of elements in a mapping, issuing ERC721 ids, or counting request ids.
 *
 * Include with `using Counters for Counters.Counter;`
 */
library Counters {
    struct Counter {
        // This variable should never be directly accessed by users of the library: interactions must be restricted to
        // the library's function. As of Solidity v0.5.2, this cannot be enforced, though there is a proposal to add
        // this feature: see https://github.com/ethereum/solidity/issues/4637
        uint256 _value; // default: 0
    }

    function current(Counter storage counter) internal view returns (uint256) {
        return counter._value;
    }

    function increment(Counter storage counter) internal {
        unchecked {
            counter._value += 1;
        }
    }

    function decrement(Counter storage counter) internal {
        uint256 value = counter._value;
        require(value > 0, "Counter: decrement overflow");
        unchecked {
            counter._value = value - 1;
        }
    }

    function reset(Counter storage counter) internal {
        counter._value = 0;
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: contracts/bingo/ENSReverseRegistration.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface IENS {
    event NewOwner(bytes32 indexed node, bytes32 indexed label, address owner);
    event Transfer(bytes32 indexed node, address owner);
    event NewResolver(bytes32 indexed node, address resolver);
    event NewTTL(bytes32 indexed node, uint64 ttl);
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    function setRecord(bytes32, address, address, uint64) external;
    function setSubnodeRecord(bytes32, bytes32, address, address, uint64) external;
    function setSubnodeOwner(bytes32, bytes32, address) external returns(bytes32);
    function setResolver(bytes32, address) external;
    function setOwner(bytes32, address) external;
    function setTTL(bytes32, uint64) external;
    function setApprovalForAll(address, bool) external;
    function owner(bytes32) external view returns (address);
    function resolver(bytes32) external view returns (address);
    function ttl(bytes32) external view returns (uint64);
    function recordExists(bytes32) external view returns (bool);
    function isApprovedForAll(address, address) external view returns (bool);
}

interface IReverseRegistrar {
    function ADDR_REVERSE_NODE() external view returns (bytes32);
    function ens() external view returns (IENS);
    function defaultResolver() external view returns (address);
    function claim(address) external returns (bytes32);
    function claimWithResolver(address, address) external returns (bytes32);
    function setName(string calldata) external returns (bytes32);
    function node(address) external pure returns (bytes32);
}

library ENSReverseRegistration {
    // namehash('addr.reverse')
    bytes32 internal constant ADDR_REVERSE_NODE = 0x91d1777781884d03a6757a803996e38de2a42967fb37eeaca72729271025a9e2;

    function setName(address ensregistry, string calldata ensname) internal {
        IReverseRegistrar(IENS(ensregistry).owner(ADDR_REVERSE_NODE)).setName(ensname);
    }
}

// File: contracts/bingo/GameFactory.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/Counters.sol";
import "@openzeppelin/contracts/security/Pausable.sol";
import "@openzeppelin/contracts/token/ERC721/IERC721.sol";
import "@openzeppelin/contracts/token/ERC721/IERC721Receiver.sol";
import "@openzeppelin/contracts/token/ERC1155/IERC1155.sol";
import "@openzeppelin/contracts/token/ERC1155/utils/ERC1155Receiver.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "./Roller.sol";
import "./ENSReverseRegistration.sol";

interface BingoCardsNFT {
    function boardArray(uint tokenId) external view returns (uint8[5][5] memory);
    function boardString(uint tokenId) external view returns (string memory);
    function ownerOf(uint256 tokenId) external view returns (address);
}

contract GameFactory is ERC1155Receiver, IERC721Receiver, Ownable, Pausable, ReentrancyGuard {
    using Counters for Counters.Counter;

    enum TokenType { NONE, ERC721, ERC1155 }

    struct Game {
        address     owner;
        address     tokenAddress;
        uint256     tokenId;
        uint256     startingBlock;
        TokenType   tokenType;
    }

    mapping(address => mapping(uint => uint)) gameIndexByOwnerAndId;
    bytes32 public constant MINTER_ROLE = keccak256("MINTER_ROLE");
    Roller  public roller;
    uint    public defaultBlockPeriod = 3;
    uint    public constant MAX_BLOCK_PERIOD = 24;
    BingoCardsNFT               public bingoCardsNFT;
    mapping(uint => Game)       public games;
    mapping (address => uint[]) public gameIdsByOwner;
    mapping(uint => address)    public winners;
    Counters.Counter            private gameCounter;
    mapping(uint => uint)       private blockPeriods;
    mapping(uint => bytes32)    public passcodes;

    event GameCreated(uint indexed gameId, address indexed owner, address indexed tokenAddress, uint256 tokenId);
    event GameDeleted(uint indexed gameId, address indexed owner, address caller);
    event Deposited(address indexed user, address indexed tokenContract, uint256 tokenId);
    event Withdrawn(address indexed user, address indexed tokenContract, uint256 tokenId);
    event BINGO(uint indexed gameId, uint tokenId, address indexed winner);

    constructor() {
        roller = Roller(0xDc582388f58CB90F9044CB866166B9921d7cF64D); // mainnet roller
        bingoCardsNFT = BingoCardsNFT(0xA99cAb29165914300C9Ec34c87a33E89C9f38769); // mainnet NFTs
    }

// NFT Transfer

    function newGameERC721(address tokenAddress, uint256 tokenId, uint rollFreq) external {
        require(IERC721(tokenAddress).ownerOf(tokenId) == msg.sender, "Not the owner.");
        require(rollFreq <= MAX_BLOCK_PERIOD, "Invalid block period");
        IERC721(tokenAddress).transferFrom(msg.sender, address(this), tokenId);
        if (rollFreq > 0 && rollFreq != defaultBlockPeriod) {
            blockPeriods[gameCounter.current()] = rollFreq;
        }   
        _newGame(msg.sender, tokenAddress, tokenId, TokenType.ERC721);
    }

    function newGameERC1155(address tokenAddress, uint256 tokenId, uint rollFreq) external {
        require(IERC1155(tokenAddress).balanceOf(msg.sender, tokenId) > 0, "Not the owner.");
        require(rollFreq <= MAX_BLOCK_PERIOD, "Invalid block period");
        IERC1155(tokenAddress).safeTransferFrom(msg.sender, address(this), tokenId, 1, "");
        if (rollFreq > 0 && rollFreq != defaultBlockPeriod) {
            blockPeriods[gameCounter.current()] = rollFreq;
        }   
        _newGame(msg.sender, tokenAddress, tokenId, TokenType.ERC1155);
    }

    function onERC721Received(address, address from, uint256 tokenId, bytes calldata) external override returns (bytes4) {
        require(msg.sender == address(IERC721(msg.sender)), "Only genuine ERC721 transfers allowed.");
        _newGame(from, msg.sender, tokenId, TokenType.ERC721);
        return this.onERC721Received.selector;
    }

    function onERC1155Received(address, address from, uint256 id, uint256 value, bytes calldata) external override returns(bytes4) {
        require(value == 1, "Can only deposit 1 token at a time.");
        require(msg.sender == address(IERC1155(msg.sender)), "Only genuine ERC1155 transfers allowed.");
        _newGame(from, msg.sender, id, TokenType.ERC1155);
        return this.onERC1155Received.selector;
    }

    function onERC1155BatchReceived(address, address from, uint256[] calldata ids, uint256[] calldata values, bytes calldata) external override returns(bytes4) {
        require(msg.sender == address(IERC1155(msg.sender)), "Only genuine ERC1155 batch transfers allowed.");
        for (uint256 i = 0; i < ids.length; i++) {
            _newGame(from, msg.sender, ids[i], TokenType.ERC1155);
        }
        return this.onERC1155BatchReceived.selector;
    }

    function withdraw(uint gameId) external nonReentrant {
        require(games[gameId].owner == msg.sender, "Not the owner.");
        Game storage _game = games[gameId];
        if (_game.tokenType == TokenType.ERC1155) {
            IERC1155(_game.tokenAddress).safeTransferFrom(address(this), msg.sender, _game.tokenId, 1, "");
        } else if (_game.tokenType == TokenType.ERC721) {
            IERC721(_game.tokenAddress).transferFrom(address(this), msg.sender, _game.tokenId);
        }
        emit Withdrawn(msg.sender, _game.tokenAddress, _game.tokenId);
        deleteGame(gameId);
    }

// Game Management

    function _newGame(address gameOwner, address tokenAddress, uint256 tokenId, TokenType tokenType) private whenNotPaused returns (uint) {
        uint currentGameId = gameCounter.current();
        games[currentGameId] = Game(gameOwner, tokenAddress, tokenId, 0, tokenType);
        uint[] storage ownerGames = gameIdsByOwner[gameOwner]; 
        uint indexInOwnerList = ownerGames.length;
        ownerGames.push(currentGameId);
        gameIndexByOwnerAndId[gameOwner][currentGameId] = indexInOwnerList; 
        gameCounter.increment();
        emit GameCreated(currentGameId, gameOwner, tokenAddress, tokenId); 
        return currentGameId; 
    }

    function deleteGame(uint _gameId) private  {
        require(_gameId < gameCounter.current(), "Invalid game ID"); 
        address _owner = games[_gameId].owner;
        uint[] storage ownerGameIds = gameIdsByOwner[_owner];
        uint indexToDelete = gameIndexByOwnerAndId[_owner][_gameId];
        require(ownerGameIds[indexToDelete] == _gameId, "Data inconsistency!");  
        ownerGameIds[indexToDelete] = ownerGameIds[ownerGameIds.length - 1];
        gameIndexByOwnerAndId[_owner][ownerGameIds[indexToDelete]] = indexToDelete;
        ownerGameIds.pop();
        emit GameDeleted(_gameId, _owner, msg.sender);
        delete gameIndexByOwnerAndId[_owner][_gameId];  // Clean up
    }

    function startGame(uint _gameId) public whenNotPaused {
        require(_gameId < gameCounter.current(), "Invalid game ID"); 
        require(games[_gameId].owner == msg.sender, "Only owner can start game");
        require(games[_gameId].startingBlock == 0 || !isActive(_gameId), "Game in progress");

        games[_gameId].startingBlock = block.number;
    }

    function listGames(address _owner) external view returns (uint[] memory) {
        return gameIdsByOwner[_owner];
    }

    function gameCount() public view returns (uint) {
        return gameCounter.current();
    }

    function gameCountByOwner(address _owner) public view returns (uint) {
        return gameIdsByOwner[_owner].length;
    }


    function setBlockFreq(uint _gameId, uint _numBlocks) public {
        require(_gameId < gameCounter.current(), "Invalid game ID"); 
        require(games[_gameId].owner == msg.sender, "Only owner can set block freq");
        require(games[_gameId].startingBlock == 0, "Game already started");
        blockPeriods[_gameId] = _numBlocks;
    }

    function getBlockPeriod(uint _gameId) public view returns (uint) {
        require(_gameId < gameCounter.current(), "Invalid game ID"); 
        return blockPeriods[_gameId] == 0 ? defaultBlockPeriod : blockPeriods[_gameId];
    }

// Passcodes

    function setPasscode(uint _gameId, bytes32 _passcode) public {
        require(_gameId < gameCounter.current(), "Invalid game ID"); 
        require(games[_gameId].owner == msg.sender, "Only owner can set passcode");
        passcodes[_gameId] = _passcode;
    }

    function checkPasscode(uint _gameId, string memory _passcode) public view returns (bool) {
        require(_gameId < gameCounter.current(), "Invalid game ID"); 
        return passcodes[_gameId] == keccak256(abi.encodePacked(_passcode));
    }

// Rolled Numbers

    function rollCounts(uint _gameId) public view returns (uint, uint) {
        require(games[_gameId].startingBlock > 0, "Game not started");
        require(_gameId < gameCounter.current(), "Invalid game ID"); 
        return rollCountsInternal(_gameId);
    }

    function rollCountsInternal(uint _gameId) internal view returns (uint, uint) {
        uint _blockPeriod = blockPeriods[_gameId] > 0 ? blockPeriods[_gameId] : defaultBlockPeriod;
        return roller.count(games[_gameId].startingBlock, _blockPeriod);
    }

    function isActive(uint _gameId) public view returns (bool) {
        (uint _rolls, uint _total) = rollCountsInternal(_gameId);
        return (_rolls <= _total);
    }

    function getBlockNumbers(uint _gameId) public view returns (uint[] memory) {
        require(_gameId < gameCounter.current(), "Invalid game ID"); 
        require(games[_gameId].startingBlock > 0, "Game not started");
        uint _blockPeriod = blockPeriods[_gameId] > 0 ? blockPeriods[_gameId] : defaultBlockPeriod;
        return roller.getBlockNumbers(games[_gameId].startingBlock, _blockPeriod);
    }

    function getRolls(uint _gameId) public view returns (uint[] memory) {
        require(_gameId < gameCounter.current(), "Invalid game ID"); 
        require(games[_gameId].startingBlock > 0, "Game not started");
        uint _blockPeriod = blockPeriods[_gameId] > 0 ? blockPeriods[_gameId] : defaultBlockPeriod;
        return roller.getRolls(games[_gameId].startingBlock, _blockPeriod, makeSeed(_gameId));
    }
    
    function currentBlockNumber() public view returns (uint) {
        return block.number;
    }

    function makeSeed(uint _gameId) internal view returns (bytes32) {
        return keccak256(abi.encodePacked(_gameId, games[_gameId].owner, games[_gameId].startingBlock));
    }

// Bingo Cards

    function boardNumbersArray(uint tokenId) external view returns (uint8[5][5] memory){
        return bingoCardsNFT.boardArray(tokenId);
    }

    function boardNumbersString(uint tokenId) external view returns (string memory){
        return bingoCardsNFT.boardString(tokenId);
    }

    function fiveFromBoard(uint tokenId, uint orientation, uint index) public view returns (uint[5] memory) {
        require(index < 5, "Invalid index");
        uint[5] memory result;
        uint8[5][5] memory board = bingoCardsNFT.boardArray(tokenId);
        if (orientation == 0) {         // HORIZONTAL
            result[0] = board[index][0];
            result[1] = board[index][1];
            result[2] = board[index][2];
            result[3] = board[index][3];
            result[4] = board[index][4];
            return result;
        } else if (orientation == 1) { // VERTICAL
            result[0] = board[0][index];
            result[1] = board[1][index];
            result[2] = board[2][index];
            result[3] = board[3][index];
            result[4] = board[4][index];
            return result;
        } else if (orientation == 2) { // DIAGONAL UP
            result[0] = board[4][0];
            result[1] = board[3][1];
            result[2] = 0;
            result[3] = board[1][3];
            result[4] = board[0][4];
            return result;
        } else if (orientation == 3) { // REVERSE DOWN
            result[0] = board[0][0];
            result[1] = board[1][1];
            result[2] = 0;
            result[3] = board[3][3];
            result[4] = board[4][4];
            return result;
        } else if (orientation == 4) { // FOUR CORNERS
            result[0] = board[0][0];
            result[1] = board[0][4];
            result[2] = 0; 
            result[3] = board[4][0];
            result[4] = board[4][4];
            return result;
        } else
            revert("Invalid orientation");
    }

    function fiveRolls(uint gameId, uint[5] memory indexes) public view returns (uint[5] memory) {
        require(isActive(gameId), "Game is not active");
        uint[5] memory result;
        uint[] memory rolls = getRolls(gameId);
        result[0] = rolls[indexes[0]];
        result[1] = rolls[indexes[1]];
        result[2] = rolls[indexes[2]];
        result[3] = rolls[indexes[3]];
        result[4] = rolls[indexes[4]];
        return result;
    }

    function fiveRollsCheap(uint gameId, uint[5] memory indexes) public view returns (uint[5] memory){
        require(isActive(gameId), "Game is not active");
        require(gameId < gameCounter.current(), "Invalid game ID"); 
        require(games[gameId].startingBlock > 0, "Game not started");
        uint _blockPeriod = blockPeriods[gameId] > 0 ? blockPeriods[gameId] : defaultBlockPeriod;
        return roller.fiveRollsByIndex(indexes, games[gameId].startingBlock, _blockPeriod, makeSeed(gameId));
    }

    //  PARAMS: 
    //  ----------------------------    
    //  gameId       game ID
    //  tokenId      gameBoard NFT
    //  orientation  0=horizontal, 1=vertical, 2=diagonal up, 3=diagonal down, 4=four corners
    //  index        0-4 to specify position of the 5 numbers on the game board
    //  indexes      5 indexes of the rolls to check

    function claimWinner(uint gameId, uint tokenId, uint orientation, uint index, uint[5] memory indexes, string memory _passcode) public whenNotPaused {
        require(passcodes[gameId] == 0 || checkPasscode(gameId, _passcode), "Invalid passcode");
        require(isActive(gameId), "Game is not active");
        uint[5] memory fromBoard = fiveFromBoard(tokenId, orientation, index);
        uint[5] memory rolls = fiveRolls(gameId, indexes);
        require(winners[gameId] == address(0), "Game already won");
        require(fromBoard[0] == rolls[0], "Roll 0 doesn't match");
        require(fromBoard[1] == rolls[1], "Roll 1 doesn't match");
        require(fromBoard[2] == rolls[2] || fromBoard[2] == 0, "Roll 2 doesn't match"); // Center can be 0
        require(fromBoard[3] == rolls[3], "Roll 3 doesn't match");
        require(fromBoard[4] == rolls[4], "Roll 4 doesn't match");
        address _owner = bingoCardsNFT.ownerOf(tokenId);
        winners[gameId] = _owner;
    
        // Send NFT to winner
        Game storage _game = games[gameId];
        if (_game.tokenType == TokenType.ERC1155) {
            IERC1155(_game.tokenAddress).safeTransferFrom(address(this), _owner, _game.tokenId, 1, "");
        } else if (_game.tokenType == TokenType.ERC721) {
            IERC721(_game.tokenAddress).transferFrom(address(this), _owner, _game.tokenId);
        }
        emit Withdrawn(_owner, _game.tokenAddress, _game.tokenId);
        emit BINGO(gameId, tokenId, _owner);
        deleteGame(gameId); // delete from users list of games  
    }

    function checkWinner(uint gameId, uint tokenId, uint orientation, uint index, uint[5] memory indexes) public view returns (bool) {
        require(isActive(gameId), "Game is not active");
        uint[5] memory fromBoard = fiveFromBoard(tokenId, orientation, index);
        uint[5] memory rolls = fiveRolls(gameId, indexes);
        require(winners[gameId] == address(0), "Game already won");
        require(fromBoard[0] == rolls[0], "Roll 0 doesn't match");
        require(fromBoard[1] == rolls[1], "Roll 1 doesn't match");
        require(fromBoard[2] == rolls[2] || fromBoard[2] == 0, "Roll 2 doesn't match"); // Center can be 0
        require(fromBoard[3] == rolls[3], "Roll 3 doesn't match");
        require(fromBoard[4] == rolls[4], "Roll 4 doesn't match");
        return true;
    }

// Contract Admin

    function pause() public onlyOwner {
        _pause();
    }

    function unpause() public onlyOwner {
        _unpause();
    }

    function setRoller(address _roller) public onlyOwner {
        require(address(roller) == address(0), "Roller already set");
        roller = Roller(_roller);
    }

    function setDefaultBlockPeriod(uint _blockPeriod) public onlyOwner {
        defaultBlockPeriod = _blockPeriod;
    }

    function setBingoCardAddr(address _bingoCardsNFT) public onlyOwner {
        bingoCardsNFT = BingoCardsNFT(_bingoCardsNFT);
    }

    function setName(address ensRegistry, string calldata ensName) external onlyOwner {
        ENSReverseRegistration.setName(ensRegistry, ensName);
    }


}
// File: contracts/bingo/Roller.sol



pragma solidity ^0.8.0;

/* 
*       ____
*      /\' .\    _____
*     /: \___\  / .  /\
*     \' / . / /____/..\
*      \/___/  \'  '\  /
*               \'__'\/                                 
*/

contract Roller {

    uint public constant MAX_VALUE = 75;
    uint public constant MAX_BLOCKS = 256;

    constructor() {
    }

    function count(uint startBlock, uint freq) public view returns (uint, uint) {
        require(startBlock <= block.number, "Start block must be less than or equal to the current block");
        require(freq > 0, "Frequency must be greater than 0");
        return (((block.number - startBlock) + freq - 1) / freq, MAX_BLOCKS / freq);
    }

    function getBlockNumbers(uint startBlock, uint freq) public view returns (uint[] memory) {
        require(startBlock <= block.number, "Start block must be less than or equal to the current block");
        require(startBlock > block.number - MAX_BLOCKS, "Block hash is not stored");
        require(freq > 0, "Frequency must be greater than 0");
        (uint arraySize,) = count(startBlock, freq);
        uint[] memory blocks = new uint[](arraySize);

        for(uint i = 0; i < arraySize; i++) {
            if (startBlock + freq * i >= block.number)
                break;
            else
                blocks[i] = startBlock + freq * i;
        }  
        return blocks;
    }

    function getRolls(uint startBlock, uint freq, bytes32 seed) public view returns (uint[] memory) {
        require(startBlock <= block.number, "Start block must be less than or equal to the current block");
        require(startBlock > block.number - MAX_BLOCKS, "Block hash is not stored");
        require(freq > 0, "Frequency must be greater than 0");
        (uint arraySize,) = count(startBlock, freq);
        uint[] memory blocks = new uint[](arraySize);

        for(uint i = 0; i < arraySize; i++) {
            if (startBlock + freq * i >= block.number)
                break;
            else
                blocks[i] = generateRandomNumber(startBlock + freq * i, seed);
        }  
        return blocks;
    }

    function fiveRollsByIndex(uint[5] memory indexes, uint startBlock, uint freq, bytes32 seed) public view returns (uint[5] memory) {
        require(startBlock <= block.number, "Start block must be less than or equal to the current block");
        require(startBlock > block.number - MAX_BLOCKS, "Block hash is not stored");
        require(freq > 0, "Frequency must be greater than 0");
        require(indexes.length == 5, "Array length must be 5");
        uint[5] memory rolls;
        for(uint i = 0; i < 5; i++) {
            if (startBlock + freq * indexes[i] >= block.number)
                break;
            else
                rolls[i] = generateRandomNumber(startBlock + freq * indexes[i], seed);
        }  
        return rolls;
    }

// Check Numbers

    function isNumberInList(uint startBlock, uint freq, uint number) public view returns (bool) {
        require(startBlock <= block.number, "Start block must be less than or equal to the current block");
        require(startBlock > block.number - MAX_BLOCKS, "Block hash is not stored");
        if (freq == 0)
            freq++;

        if (number >= startBlock && number < block.number) {
            if ((number - startBlock) % freq == 0) {
                return true;
            }
        }
        return false;
    }

    function areNumbersInList(uint startBlock, uint freq, uint[] memory numbers) public view returns (bool) {
        require(numbers.length >= 1 && numbers.length <= 5, "Array length must be between 1 and 5");

        for (uint i = 0; i < numbers.length; i++) {
            if (!isNumberInList(startBlock, freq, numbers[i])) {
                return false;
            }
        }
        return true;
    }

// Misc

    function currentBlockNumber() public view returns (uint) {
        return block.number;
    }

// Internal

    function generateRandomNumber(uint256 _blockNumber, bytes32 _seed) internal view returns (uint256) {
        require(block.number > _blockNumber, "Block not yet mined");
        require(_blockNumber > block.number - MAX_BLOCKS, "Block hash is not stored");
        return uint256(keccak256(abi.encodePacked(blockhash(_blockNumber), _seed))) % MAX_VALUE + 1;
    }

}